Three ways to create vectors

```
> u := Vector([1,2]);
```

$$u := \begin{bmatrix} 1 \\ 2 \end{bmatrix}$$

```
> v := <2,3>;
```

$$v := \begin{bmatrix} 2 \\ 3 \end{bmatrix}$$

```
> w := Vector(3);
```

$$w := \begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix}$$

```
> w[1] := 2;
```

$$w_1 := 2$$

```
> for i to 3 do w[i] := i^2; od:
  w;
```

$$\begin{bmatrix} 1 \\ 4 \\ 9 \end{bmatrix}$$

Vector addition, scalar multiplication and dot product

```
> u+v;
```

$$\begin{bmatrix} 3 \\ 5 \end{bmatrix}$$

```
> 2*v;
```

$$\begin{bmatrix} 4 \\ 6 \end{bmatrix}$$

```
> u.v;
```

$$8$$

Two ways to input matrices

```
> A := Matrix([[1,1],[1,0]]);
```

$$A := \begin{bmatrix} 1 & 1 \\ 1 & 0 \end{bmatrix}$$

```
> B := Matrix(2,2);
```

$$B := \begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix}$$

```
> B[1,1] := 1;
  B[1,2] := 2;
```

$$B_{1,1} := 1$$
$$B_{1,2} := 2$$

```
> for i to 2 do B[2,i] := i^2; od;
```

$$B_{2,1} := 1$$
$$B_{2,2} := 4$$

```
> B;
```

$$\begin{bmatrix} 1 & 2 \\ 1 & 4 \end{bmatrix}$$

Matrix addition

```
> A+B;
```

$$\begin{bmatrix} 2 & 3 \\ 2 & 4 \end{bmatrix}$$

Matrix multiplication - is not commutative in general

```
> A.B;
```

$$\begin{bmatrix} 2 & 6 \\ 1 & 2 \end{bmatrix}$$

```
> B.A;
```

$$\begin{bmatrix} 3 & 1 \\ 5 & 1 \end{bmatrix}$$

```
> 2*A;
```

$$\begin{bmatrix} 2 & 2 \\ 2 & 0 \end{bmatrix}$$

Matrix inverse

```
> 1/A;
```

$$\begin{bmatrix} 0 & 1 \\ 1 & -1 \end{bmatrix}$$

Matrix vector multiplication

```
> A.u;
```

$$\begin{bmatrix} 3 \\ 1 \end{bmatrix}$$

The LinearAlgebra package
We'll look at the Determinant, LinearSolve, ReducedRowEchelonForm, CharacteristicPolynomial, Eigenvalues, Eigenvectors commands

```
> with(LinearAlgebra):
```

Solve $A \cdot x = u$. Two ways.

```
> A^(-1).u;
```

$$\begin{bmatrix} 2 \\ -1 \end{bmatrix}$$

```
> x := LinearSolve(A,u);
```

$$x := \begin{bmatrix} 2 \\ -1 \end{bmatrix}$$

```
> A.x = u;
```

$$\begin{bmatrix} 1 \\ 2 \end{bmatrix} = \begin{bmatrix} 1 \\ 2 \end{bmatrix}$$

```
> A;
```

$$\begin{bmatrix} 1 & 1 \\ 1 & 0 \end{bmatrix}$$

```
> Determinant(A);
```

$$-1$$

```
> I2 := IdentityMatrix(2);
```

$$I2 := \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$$

```
> B := <A|I2>;
```

$$B := \begin{bmatrix} 1 & 1 & 1 & 0 \\ 1 & 0 & 0 & 1 \end{bmatrix}$$

```
> C := ReducedRowEchelonForm(B);
```

$$C := \begin{bmatrix} 1 & 0 & 0 & 1 \\ 0 & 1 & 1 & -1 \end{bmatrix}$$

```
> C[1..2,3..4];
```

$$\begin{bmatrix} 0 & 1 \\ 1 & -1 \end{bmatrix}$$

```
> 1/A;
```

$$\begin{bmatrix} 0 & 1 \\ 1 & -1 \end{bmatrix}$$

```
> C := CharacteristicPolynomial(A,lambda);
```

$$C := \lambda^2 - \lambda - 1$$

```
> solve(C=0,lambda);
```

$$\frac{1}{2}\sqrt{5} + \frac{1}{2}, \frac{1}{2} - \frac{1}{2}\sqrt{5}$$

```
> Eigenvalues(A);
```

$$\begin{bmatrix} \frac{1}{2}\sqrt{5}+\frac{1}{2} \\ \frac{1}{2}-\frac{1}{2}\sqrt{5} \end{bmatrix}$$

```
> E := Eigenvalues(evalf(A));
```

$$E := \begin{bmatrix} 1.61803398874989 + 0.\,I \\ -0.618033988749895 + 0.\,I \end{bmatrix}$$

```
> E := map( Re, E);
```

$$E := \begin{bmatrix} 1.61803398874989 \\ -0.618033988749895 \end{bmatrix}$$

Create a visual for the eigenvectors and eigenvalues

```
> with(plots):
> arrow( <1,0>, scaling=constrained );
```


```
> arrow( <1,0>, <1,1>, view=[0..2,0..2] );
```




```
> A := Matrix([[1,2],[2,1]]);
```

$$A := \begin{bmatrix} 1 & 2 \\ 2 & 1 \end{bmatrix}$$

```
> n := 25:
  for i from 1 to n do
      theta := evalf(2*i*Pi/n);
```

```
    u := <cos(theta),sin(theta)>;
    unitarrow[i] := arrow(u,color=yellow);
  od:
```

```
> frames := [seq(unitarrow[i],i=1..n)]:
  display(frames);
```




Edit the above code to add code for image vectors v := A.u

```
> n := 25:
  for i from 1 to n do
     theta := evalf(2*i*Pi/n);
     u := <cos(theta),sin(theta)>;
     unitarrow[i] := arrow(u,color=yellow);
     v := A.u;
     imagearrow[i] := arrow(u,v,color=red);
  od:
```

```
> frames := [seq(unitarrow[i],i=1..n), seq(imagearrow[i],i=1..n)]:
  display(frames,scaling=constrained);
```




Can you see that the eigenvectors are <1,1> and <1,-1> here?

```
> A;
```

$$\begin{bmatrix} 1 & 2 \\ 2 & 1 \end{bmatrix}$$

```
> Eigenvectors(A);
```

$$\begin{bmatrix} -1 \\ 3 \end{bmatrix}, \begin{bmatrix} -1 & 1 \\ 1 & 1 \end{bmatrix}$$

The columns of the matrix are the eigenvectors. So <-1,1> and <1,1> are the eigenvectors.

Finally, edit the above code to display the unitvector u and image vector v together as a single frame.

```
> n := 61;
  for i from 1 to n do
     theta := evalf(2*i*Pi/n);
     u := <cos(theta),sin(theta)>;
     v := A.u;
     unitarrow[i] := arrow( u, color=yellow );
     imagearrow[i] := arrow( u, v, color=red );
     frame[i] := display( [unitarrow[i],imagearrow[i]] );
  od:
```

$$n := 61$$

```
> frames := [seq(frame[i],i=1..n)]:
  display(frames,insequence=true);
```




The power method for computing an eigenvectors
Start with a random vector v and compute A^n v

```
> v := <1.0,0.0>;
  to 10 do
     v := A.v;
     d := sqrt(v[1]^2+v[2]^2);
     v := v/d;
  od:
  v;
```

$$v := \begin{bmatrix} 1.0 \\ 0. \end{bmatrix}$$

$$\begin{bmatrix} 0.707118756000577 \\ 0.707094806169722 \end{bmatrix}$$

Try u = <0,1>, <0,-1>, <-1,0>
You should find that it always converges to the eigenvector <1,1>.
How can we make it converge to the other eigenvector <-1,1>?

Matrices and Vectors in Maple can have symbolic entries

```
> A := Matrix([[a,b],[b,a]]);
```

$$A := \begin{bmatrix} a & b \\ b & a \end{bmatrix}$$

```
> Determinant(A);
```

$$a^2 - b^2$$

```
> 1/A;
```

$$\begin{bmatrix} \frac{a}{a^2 - b^2} & -\frac{b}{a^2 - b^2} \\ -\frac{b}{a^2 - b^2} & \frac{a}{a^2 - b^2} \end{bmatrix}$$